

Másik hasznos billentyűzettel kapcsolatos függvény a **keypressed**. Ennek értéke igaz/hamis lehet. Értéke akkor igaz, ha a felhasználó lenyomott valamilyen billentyűt. Ilyenkor a lenyomott billentyűt a **readkey** segítségével olvashatjuk ki. A következő program addig fogja a "Hello!" szót kiírni a képernyőre véletlenszám generátorral kigenerált helyre véletlen színnel, amíg nem nyomunk meg egy billentyűt.

```
program Pelda43;
uses crt;
var c:char;
begin
  randomize;
  clrscr;
  repeat
    gotoxy (random (75)+1, random (24)+1) ;
    textcolor (random (15)+1) ;
    write ('Hello! ');
    delay (100) ;
  until keypressed;
  c:=readkey;
end.
```

A **delay(100)** parancs leállítja 100 millisekundumra a program futását, tehát minden kiírás után vár 100 msec-ot. Ennek a parancsnak a segítségével tudjuk a fenti programunkban megadni, hogy milyen gyorsan írja ki a program a képernyőre az üzeneteket.

A következő program kiírja egy lenyomott billentyű ASCII kódját. Ez a program hasznos segédprogram lehet a későbbiekben, ha meg szeretnénk határozni melyik billentyűhöz milyen ASCII kód tartozik. Az ESC kódja a 27, ezért van a ciklus feltételénél a **c=#27** kifejezés. A ciklus tehát akkor fejeződik be, ha a c-ben levő karakter ASCII kódja egyenlő 27-tel, vagyis ha az ESC billentyűt nyomtuk meg.

```
program Pelda44;
uses crt;
var c:char;
begin
  clrscr;
  repeat
    c:=readkey;
    writeln(c, ' kódja: ', ord(c));
  until c=#27;
end.
```

Ennek a programnak a segítségével figyeljük meg az egyes billentyűk ASCII kódjait. Észrevehetjük, hogy némelyik billentyűnél - pl. F1, F2, ... és a nyilak lenyomásánál két kód jelenik meg. Ez azért van, mivel ezeknél a gomboknál a billentyűzet két kódot küld a számítógépnek, először a 0-t, majd egy másik kódot.

Feladat: Készíts programot melyben induláskor kék háttérrel látsz, majd ENTER lenyomására rajzol egy zöld téglalapot (10,10,20,20), melybe sárga betűkkel legyen beleírva a „Teglalap” szöveg!

```
program Pelda45;
uses ctr;
begin
  textbackground (blue) ;
  clrscr;
  readln;
  textcolor (14) ;
  window (10,10,20,20) ;
  textbackground (green) ;
  clrscr;
  writeln ('Teglalap');
  readln;
end.
```

Az alábbi program (következő oldal) bemutatja hogyan készíthetünk egy egyszerű menüt az eddig tanultak alapján, melyben a felfelé ill. lefelé nyilak segítségével mozoghatunk, Enter-rel aktiválhatjuk a kiválasztott menüpontot és Esc-el léphetünk ki a programból (a későbbiekben sokszor felhasználhatjuk ezt a programot):

```

program Pelda46;
uses crt;
var k:integer; { ebben fogjuk tarolni, hogy éppen melyik menuponton vagyunk }
    c:char;
begin
  clrscr;
  k:=1;
  { kiirja a menut }
  textbackground(red);
  textcolor(white);
  gotoxy(10,10);
  write(' Elso ');
  textbackground(blue);
  textcolor(yellow);
  gotoxy(10,11);
  write(' Masodik ');
  gotoxy(10,12);
  write(' Harmadik ');
  { a kurzort beallitjuk a jobb also sarokba }
  gotoxy(80,25);
  { beolvas egy billentyut es a menut ettol fuggoen atrajzolja }
  repeat
    c:=readkey;
    { ha valamelyik nyil lett megnyomva }
    if c=#0 then begin
      { atfestjuk a kivlasztottat kekre }
      textbackground(blue);
      textcolor(yellow);
      gotoxy(10,9+k);
      case k of
        1: write(' Elso ');
        2: write(' Masodik ');
        3: write(' Harmadik ');
      end;
      { megnezzuk melyik billentyut nyomta meg a felhasznalo
        es ettol fuggoen megvaltoztatjuk a k erteket }
      c:=readkey;
      case c of
        #72: if k>1 then dec(k); { #72 = felfele nyil }
        #80: if k<3 then inc(k); { #80 = lefele nyil }
      end;
      { atfestjuk az uj kivlasztottat pirosra }
      textbackground(red);
      textcolor(white);
      gotoxy(10,9+k);
      case k of
        1: write(' Elso ');
        2: write(' Masodik ');
        3: write(' Harmadik ');
      end;
      { a kurzort beallitjuk a jobb also sarokba }
      gotoxy(80,25);
      end;
    { ha #13 = Enter lett megnyomva }
    if c=#13 then begin
      gotoxy(10,15);
      textbackground(0);
      textcolor(7);
      writeln('Kivalaszottad a(z) ',k,'. menupontot!');
      gotoxy(80,25);
      end;
  until c=#27;
end.

```

A crt unit segítségével hangokat is adhatunk ki. Erre a **sound**(*frekvencia*) ill. a **nosound** parancsok szolgálnak. Pl.

```

program Pelda47;
uses crt;
begin
  clrscr;
  sound(800);
  delay(1000);
  sound(1200);
  delay(500);
  sound(1000);
  delay(500);
  nosound;
  readln;
end.

```

Feladat: Mozgassunk egy téglalapot (egy kis képernyőt) benne egy szöveggel a képernyőn a kurzormozgató billentyűk segítségével!

```

program Pelda48;
uses Crt;
var  x1, x2, y1, y2: byte;
     c: char;
begin
  x1:=35; y1:=12; x2:=45; y2:=16;
  repeat
    {A régi ablak törlése}
    TextBackground(black);
    ClrScr;
    {Új ablak és a szöveg megjelenítése}
    Window(x1, y1, x2, y2);
    TextBackground(blue);
    TextColor(red);
    ClrScr;
    GotoXY(3, 3);
    WriteLn('szoveg');
    {Várakozás egy billentyű leütésére}
    c := ReadKey;
    {Ha a billentyűzetnek két bájtos kódja van (az első bájtt #0), a második bájtt
    beolvasása.
    Ilyenek a kurzormozgató billentyűk.}
    if c = #0 then
      begin
        c := ReadKey;
        case c of
          #72: begin Dec(y1); Dec(y2) end; {Felfele nyíl}
          #80: begin Inc(y1); Inc(y2) end; {Lefele nyíl}
          #77: begin Inc(x1); Inc(x2) end; {Jobbra nyíl}
          #75: begin Dec(x1); Dec(x2) end; {Balra nyíl}
        end;
      end
    end
    {Kilépés ESC-re}
  until c = #27;
  NormVideo;           {Eredeti színek visszaállítása}
  ClrScr;
end.

```